
WHAT is a system name/token pair? Simply stated,
it is an area of storage that is managed by

OS/390 services and contains a one- to 16-byte name with a
one- to 16-byte token. The name can contain characters,
integers, blanks, or an address. The token can contain any 16
bytes of information that you want to associate with the
provided name. The token is accessed by specifying the
name to be retrieved. The beauty of this is that OS/390
manages the storage applications and access, allowing the
programmer to concentrate on just using the information.

This article examines the system level usage of the
name/token pair as a repository of job accounting information
that is accessed by IEFUJV for validation purposes. There are
three other levels available (task level, home address space,
and primary address space); however, these levels are more
limited and don’t provide the functionality that was needed at
my site.

System tokens have been available at least since OS/390
Version 1 Release 1. Connect to the OS/390 Internet Library
web site (www.s390.ibm.com/os390/bkserv) and search for
IEANTCR. You will get a match on 20 different publications if
you search on the OS/390 V2R10 library. The IBM publications
are excellent resources for coding information.

THE PROJECT

The reason that I was looking for the system name/token
pairs was to solve a problem at my site. That problem involved
a very sophisticated technique for validating correct job
accounting. The technique used a table within a load module
in the linklist that was loaded and then deleted by IEFUJV for
every job. To update this table, the assembler source had to be
updated, assembled, and linkedited into a linklist library, and
an LLA update had to be performed to make the updates
available. On the bright side, we now have the ability to do an

LLA update because when this was written we had to wait for
an IPL. One downside to this approach is that the linklist
library needs to be compressed occasionally, which we only
discovered after the linkedit failed. Another downside is the
overhead of the LOAD and DELETE for the table load module
from the linklist. The code in IEFUJV to search the table was
OK, but the code could have been improved with a binary
search; however, that overhead is small compared to the
LOAD/DELETE overhead, so we decided not to bother with
it. It was time to move into the 21st Century, or at least the late
1990s. The name/token pair processing routines reside in LPA
so the overhead of the LOAD/DELETE is minimal. And,
since the name/token information also resides in memory,
there will be no DASD I/O to slow things down.

The logic that we used was much simpler and more
dynamic than the logic used in the previous technique:

1. Create and maintain a sequential file or PDS member
with the valid accounting information.

2. Write a program (started task) to:

● read each record
● delete the name/token associated with the account code

3. Create the name/token associated for the account code
4. Update IEFUJV to do a Name Get for the job’s

account code:

● If the requested account token is present, then validate
the information in the token.

● If the requested account token is not present, the
code is invalid.

By coding the name/token pair create/update program, the
program can run as a started task immediately after IPL under

SUB=MSTR, as well as whenever there is a
need to update or add new account codes.
The dynamic update can include just the
new or changed codes or a complete refresh.

However, before modifying IEFUJV I
needed to verify that I understood the
name/token pair processing. To verify my
understanding of this process, I wrote two
sample programs, TOKCREAT and
TOKGET, as shown in Figures 1 and 2. The
JCL used to test these programs is shown in
Figures 3 and 4.

Could this process be any easier? Not real-
ly. And the beauty is that OS/390 will manage
the storage of our account table and will do
the search for us. All we have to do is to cre-
ate the name/token pairs and ask OS/390 to
tell us if the account code (the name) is there.
Note: This process could work for almost
any application or product that needs to
have a customizable set of options estab-
lished before the application or product is
used. (Wouldn’t it be nice if the ISPF con-
figuration options were managed this way?)

Now let’s examine the sample code. For
the implementation for job accounting the
TOKCREAT will remain mostly unchanged.
The TOKGET code will be changed to bet-
ter integrate into IEFUJV.

GETTING STARTED

First, I went to the IBM OS/390 Internet
Library (www.s390.ibm.com/os390/bkserv)
and searched for “Token” and then narrowed
it down with a search for “IEANCTR SYS-
TEM.” The search yielded three publica-
tions for my review. In addition, section
83.1.11 of the book, OS/390 MVS Pro-
gramming: Assembler Services Reference
Document Number GC28-1910-07, con-
tains a great example of the TOKCREAT
logic that I used in my programming.

THE TOKCREAT PROGRAM

This program will create the name/token
pairs that will be accessed by the TOKGET
program. The code is shown in Figure 1.

● Statement 1 forces the program to be
linkedited as authorized. Statements 2
through 6 are the initialization and base
register set up. Statements 8 through 11
are where entry point addresses for the
Token Create routine (IEANTCR) and
the Token Delete routine (IEANTDL)

1. PUNCH ‘ SETCODE AC(1) ‘            
2. TOKCREAT CSECT            
3. TOKCREAT AMODE 31         
4. TOKCREAT RMODE 24         
5. BAKR  R14,R0     
6. LR    R12,R15    
7. USING TOKCREAT,R12                 
8. LOAD EP=IEANTDL 
9. ST R0,IEANTDL 
10. LOAD EP=IEANTCR 
11. ST R0,IEANTCR 
12. OPEN TOKIN      
13. GETTOKEN GET TOKIN      
14. LR R5,R1      
15. CLI 0(R5),C’*’ 
16. BE GETTOKEN   
17. MVC NAME,0(R5) 
18. MVC TOKEN,19(R5)                 
19. MVC WTONAME,NAME                 
20. MVC WTOTOKEN,TOKEN               
21. MODESET KEY=ZERO,MODE=SUP          
22. L R15,IEANTDL
23. CALL (15),(LEVEL,NAME,RETCODE)    
24. L R3,RETCODE 
25. C R3,=F’04’  
26. BNH CREATE     
27. L R8,RETCODE 
28. LR R9,R15     
29. ABEND 1          
30. CREATE DS 0H         
31. L R15,IEANTCR
32. CALL (15),(LEVEL,NAME,TOKEN,PERSOPT,RETCODE)        
33. MODESET KEY=NZERO,MODE=PROB        
34. CLC RETCODE,=F’0’                
35. BNE ABEND2     
36. LA R1,WTOA    
37. SVC 35         
38. B GETTOKEN   
39. ABEND2 DS 0H         
40. L R8,RETCODE 
41. LR R9,R15     
42. ABEND 2          
43. EXIT DS 0H         
44. CLOSE (TOKIN)    
45. DELETE EP=IEANTCR
46. DELETE EP=IEANDTL
47. SR R15,R15    
48. PR               
49. EJECT            
50. YREGS ,          
51. EJECT            
52. IEANTASM         
53. LTORG            
54. IEANTDL DS F          
55. IEANTCR DS F          
56. LEVEL DC A(IEANT_SYSTEM_LEVEL)        
57. NAME DS CL16       
58. TOKEN DS CL16       
59. PERSOPT DC A(IEANT_PERSIST)             
60. RETCODE DS F          
61. ORG ,          
62. DS 0F         
63. WTOA DC AL2(WTOAE-WTOA),AL2(0)       
64. WTOMSG DC C’TOKEN CREATION:  ‘         
65. WTONAME DS CL16       
66. DC C’VALUE: ‘ 
67. WTOTOKEN DS CL16       
68. WTOAE EQU *          
69. DS 0F         
70. TOKIN DCB DSORG=PS,RECFM=FB,LRECL=80,      X
71. MACRF=GL,DDNAME=TOKIN,         X
72. EODAD=EXIT 
73. END ,

FIGURE 1: SAMPLE PROGRAM TOKCREAT 

are preloaded and saved for future use.
Note: These modules reside in LPA.

● Statement 13 is the start of the
routine to read the name/token pair
information from the input card
images. The card image is read using
the Get Locate process with the
address of the current record returned
in register 1, which we then copied
into register 5 in case we decide to
insert code later to alter R1. A test is
made for a comment record as
indicated by an “*” in column 1,
if so then that record is ignored.

● Statements 17 through 20 move the input
information to the parameter lists for the
name/token processing and for the WTO
that will be issued if successful.

● Statement 21 changes the program to
Supervisor State and Key Zero which is
required for using the system level
name/token pair processes.

● Statements 22 through 29 are used to
delete the currently requested name if it
exists. The call to the IEANTDL passes
the level (of system), the name as
provided in the current input record,
and a fullword for the return code.

● Statements 30 through 38 are used to
create the new name/token pair. The
call in statement 32 passes the level
(of system), the name as provided in
the input record, the token value as
provided in the input record, a persis-
tence code (persist in this case so it will
stay around after this program ends),
and a fullword for the return code.

● Statements 36 and 37 issue a WTO if
the create was successful, thus providing
evidence in the system log, as well as
in the log for the started task/job of
what was created.

● Statements 39 through 42 are executed
if the name/token pair creation failed.

● Statements 43 through 48 are the clean
up and exit for the program.

● Statement 50 is the YREGS macro that
maps the registers so that I can code R15
instead of 15 to reference the registers.

● Statement 52 is the IEANTASM
macro, which provides the necessary
equates for easily mapping to the
name/token pair parameter lists.

● Statement 53 causes any literals used in
the program to be generated at this
point in the program.

● Statements 54 and 55 are where the
entry point addresses for the

name/token pair create and delete
routines are stored.

● Statements 56 to 60 are the information
passed to the name/token pair create
and/or delete routines.

● Statements 63 to 68 are the WTO
parameter lists and data. Statements 70
to 72 are the DCB definition for the
input file.

TECHNICAL SUPPORT • DECEMBER 2000

1. TOKGET CSECT          
2. TOKGET AMODE 31       
3. TOKGET RMODE 24       
4. BAKR R14,R0   
5. LR R12,R15  
6. USING TOKGET,R12             
7. LOAD EP=IEANTRT             
8. ST R0,IEANTRT             
9. OPEN TOKIN    
10. GETTOKEN GET TOKIN    
11. LR R5,R1    
12. MVC NAME,0(R5)             
13. L R15,IEANTRT            
14. CALL (15),(LEVEL,NAME,TOKEN,PERSOPT,RETCODE)            
15. MVC WTONAME,NAME           
16. MVC WTOTOKEN,TOKEN         
17. LA R1,WTOA  
18. SVC 35       
19. B GETTOKEN 
20. EXIT DS 0H       
21. CLOSE TOKIN    
22. DELETE EP=IEANTRT            
23. PR             
24. EJECT          
25. YREGS          
26. EJECT          
27. IEANTASM       
28. LTORG          
29. IEANTRT DS F        
30. LEVEL DC A(IEANT_SYSTEM_LEVEL)  
31. NAME DS CL16     
32. TOKEN DS CL16     
33. PERSOPT DC A(IEANT_PERSIST)       
34. RETCODE DS F        
35. WTOA DC AL2(WTOE-WTOA),AL2(0)  
36. DC C’TOKEN NAME: ‘        
37. WTONAME DC CL16’ ‘  
38. DC C’  ‘    
39. DC C’TOKEN VALUE: ‘       
40. WTOTOKEN DC CL16’ ‘  
41. WTOE EQU *        
42. TOKIN DCB DSORG=PS,RECFM=FB,LRECL=80,DDNAME=TOKIN,EODAD=EXIT,    XXXX 
43. MACRF=GL 
44. END ,

FIGURE 2: SAMPLE PROGRAM TOKGET 

//TOKCREAT JOB  SYSTEMS,’LIONEL DYCK’,CLASS=V,NOTIFY=&SYSUID, 
// MSGLEVEL=(1,1),MSGCLASS=X         
//HOLD OUTPUT JESDS=ALL,DEFAULT=Y,OUTDISP=(HOLD,HOLD) 
//CODE EXEC PGM=TOKCREAT,REGION=2M     
//STEPLIB DD  DISP=SHR,DSN=SYSX.AUTHTEST.LOAD   
//TOKIN DD  *  
*                  * START THE TOKEN VALUE IN COLUMN 20
A#.CODE1            SYSTEMS PROG

FIGURE 3: TOKCREAT STARTED TASK JCL 

//TOKGET JOB  SYSTEMS,’LIONEL DYCK’,CLASS=V,NOTIFY=&SYSUID, 
// MSGLEVEL=(1,1),MSGCLASS=X     
//HOLD OUTPUT JESDS=ALL,DEFAULT=Y,OUTDISP=(HOLD,HOLD) 
//GETTOK EXEC PGM=TOKGET,REGION=2M  
//STEPLIB DD  DISP=SHR,DSN=SYSX.TEST.LOAD  
//TOKIN DD  *    
A#.CODE00
A#.CODE1

FIGURE 4: TOKGET JCL 

THE TOKGET PROGRAM

This simple program, as shown in Figure 2,
just reads in a file with the names of the
name/token pairs that we want to display.
The TOKGET program does not have to be
authorized, but it does have to be AMODE
of 31. The call to the name/token pair get
(IEANTRT) routine in statement 14 must
tell the routine that it is looking for a name
that is at the system level and persistent.
The name is passed along with a location to
return the matching token.

Once the get request has been satisfied,
the results are displayed using a WTO in
statements 15 through 18.

THE JCL

The JCL in Figure 3 demonstrates how
to create a name/token pair with the name
A#.CODE1. Note that the name can be any
combination of characters, integers,
blanks, or an address. The A#. is used to
designate that these are accounting names,
and CODE1 is an account code id. The
token value of SYSTEMS PROG is used as
the name for the group that owns that
account code.

The JCL in Figure 4 demonstrates how to
obtain two name/token pairs by name. The
request for A#.CODE0 will cause an error
because this name does not exist. The next
request will be successful.

See Figures 5 and 6 for snippets from the
JESMSGLG for the jobs in Figures 3 and 4.

CONCLUSION AND COMMENTS

Obviously the coding for test routines could
be improved with mandatory comments. The
use of WTO could be eliminated for the
TOKGET and changed to output to a
SYSOUT if desired. The code for the
TOKGET also needs to be changed so that
it will be reentrant as required for SMF exits
and thus work within IEFUJV.

Remember that these examples are
intended to demonstrate the process only
and perform no other useful or productive
function. I am confident that you can see the
usefulness of this code for accounting vali-
dation as well as other uses for information
that needs to be available during execution
without the overhead of having to allocate
and read a DASD file. If you are a software

developer, then this technique can be used
in lieu of providing a load module for your
customer to modify for local customizations,
and it is faster than providing a parmlib set
up for local custom settings.

NaSPA member Lionel B. Dyck is a lead MVS
systems programmer for a large HMO in
California. He has been in systems programming
since 1972 and has written numerous ISPF
dialogs over the years. He is an active member
of NaSPA and SHARE, and can be reached via
email at Lionel.B.Dyck@kp.org.

IEF403I TOKCREAT - STARTED - TIME=15.26.28            
TOKEN CREATION:  A#.CODE1        VALUE: SYSTEMS PROG  
ACTRT01I TOKCREAT STEP CODE               CODE =    00
IEF404I TOKCREAT - ENDED - TIME=15.26.28

FIGURE 5: RESULTS FROM TOKCREAT JOB

IEF403I TOKGET - STARTED - TIME=15.26.28                 
+TOKEN NAME: A#.CODE0          TOKEN VALUE:              
+TOKEN NAME: A#.CODE1          TOKEN VALUE: SYSTEMS PROG 
ACTRT01I TOKGET   STEP GETTOK             CODE =    00   
IEF404I TOKGET - ENDED - TIME=15.26.28

FIGURE 6: RESULTS FROM TOKGET JOB 


